
Introduction to
ActiveX Data

Objects

Objectives

• Understand why we use ActiveX Data Objects (ADO).

• Explore the ADO object model.

• Show how to open a Recordset and work with the data it contains.

Introduction to ADO

There are many object models available to you from VBA in addition to the
ones we have discussed so far.  However, not all of them belong to VBA hosts.
One example of an object model without an associated VBA host is ADO.  In
fact, ADO doesn’t even have an associated user interface.

ADO stands for ActiveX Data Objectsthe new standard object model for
data access letting you work with data in a database through code. Why would
you want to do this? You may have a database containing names and addresses
that you want to use for a mail merge with Word or, you may want to analyze
data in a database with Microsoft Excel. There are thousands of uses for
databases and for using VBA applications to programmatically control
databases.

The intent of ADO is to offer you the same kind of access to your data, no
matter where it is.  Data could reside in Microsoft Access, another desktop
database like dBASE, FoxPro, or Paradox, or a client server database like
Microsoft SQL Server or Oracle.

For this course, we’ll be working only with .mdb files created in Microsoft
Access (Jet databases). You can consult the ADO Help file for information on
accessing other types of data.

NOTE ADO is a new and quickly changing technology; for this course
we’ll use ADO to communicate with our Access database through
the Access ODBC driver.  An OLE DB provider for Microsoft
Access may be available by the time you read this to further
optimize the communication process.

What Can You Do with ADO?

ADO gives you control over data from VBAto:

• Add tables and restructure them.

• Create and run queries.

• Read, add, edit, or delete individual records.

Setting a Reference

Before you can use ADO in your application, you must set a reference to the
Microsoft ActiveX Data Objects 1.5 Object Library, from the

Tools|References dialog box. This is true for any application. Unless you set
the reference, your application will be unable to use any of the ADO objects
and their properties or methods.

NOTE You must also set this reference for each VBA project using ADO.
Setting the reference for one project in Microsoft Excel, for
example, won’t automatically set it for other Microsoft Excel
projects.

The ADO Object Model

The ADO object model is smaller than ones you’ve seen before in this class, as
shown in Figure 1, and also much smaller than the DAO object model
previously used for getting to data in Microsoft Access databases.  Most
objects can be created without first creating a reference to the item above it in
the hierarchy. This simplifies your code and cuts down on the number of
objects to be tracked and eventually closed.

Figure 1. The ADO hierarchy is less strict than others discussed in this class.

In this section, we’ll touch on the objects you’re most likely to use in your
VBA applications.

The Connection Object

The Connection object handles communication between your application and
the data source.  In order to work with data, a connection must be either
explicitly or implicitly opened.

You can explicitly open a connection using the Open method of the
Connection object. The following code opens a connection to a Microsoft
Access database named X.MDB:

    Dim conADO As ADODB.Connection

    Set conADO = New ADODB.Connection

    conADO.Open "DRIVER=Microsoft Access Driver " & _

        "(*.mdb);" & _

        "DBQ=C:\My Documents\x.mdb"

The argument used with the Open method of the connection tells ADO what
driver to use to control communications and what database should be opened.
The argument could be amended to include a user name and password,  a Data
Set Name (DSN), or an OLE DB provider, if one has been written for the type
of data you want to use.

See the ADO help file or the OLE DB web site for more information on
specific data sources.

Getting to Data with Recordsets

Introducing the Recordset Object

In order to work programmatically with a set of rows, you must have some
way to refer to those rows as a group. ADO provides an object you’ll use any
time you work with a set of rows: the Recordset object. Simply put, this object
refers to the set of rows returned when you open a table or a query.

ADO provides four distinct types of cursors which determine how the
Recordset behaves: Dynamic, Keyset, Static, and Forward-Only. The default is
Forward-Only, but for this course we’ll focus on the Keyset cursor—the most
common and flexible type.

Key Term

Recordset An object that represents a set of rows returned by a select

query. This object is the basis of working with data

programmatically, and will figure prominently in any code you

write to manipulate data.

Keyset A type of cursor that allows the Recordset to be updated but

doesn’t immediately show the results of other changes in a

multi-user environment. This cursor type creates functionality

identical to dynasets in Microsoft Access.

Creating a Recordset

To work with data, create a new Recordset object, and use the Set statement to
make your object variable refer to a Recordset. For example, you can open a
Recordset using code like this:

Dim rst As ADODB.Recordset

Set rst = New Recordset

rst.Open "SELECT * FROM tblCustomer", _

      "DRIVER=Microsoft Access Driver (*.mdb);" & _

      "DBQ=C:\My Documents\vba.mdb;", _

      adOpenKeyset, adLockPessimistic

' Do stuff with the open Recordset.

' Close is a method of a Recordset.

' Guess what it does?

rst.Close

The Open Method

The syntax for the Open method looks like this:

var.Open Source, [ActiveConnection], _

[CursorType], [LockType], [Options]

In this syntax:

• var is a variable declared to be of type ADODB.Recordset.

• source is a string representing the source of the data (a table name,  a
SQL string, or a stored procedure).

• ActiveConnection is the object variable representing an already-open
connection or a string telling ADO how to open the connection.  In
other words, the Recordset Open method can implicitly open a
connection.

• CursorType can be one of the constants adOpenForwardOnly,
adOpenKeyset, adOpenDynamic, or adOpenStatic, indicating the type
of cursor to open.

• LockType can be one of the constants adLockReadOnly,
adLockPessimistic, adLockOptimistic, or adLockBatchOptimistic,
indicating whether editing is allowed and, if so, how locking should be
done.

• options tells ADO how to interpret source if it is something other than
a text command (i.e., SQL statement).

Using Data from the Recordset

Once you’ve created a Recordset object, you can refer to any of its properties,
methods, and collections in your code. As you saw in Figure 1, each Recordset
you create contains a collection of fields. You can loop through the fields using
the For Each...Next command structure, or you can refer to each field
individually by name or number. The following example opens a Recordset,
and then lists the values of each field for the first row.

Sub OpenARecordset()

     Dim rst As ADODB.Recordset

     Set rst = New Recordset

    ' Open a recordset on tblCustomer in

 ' vba.mdb using the Access ODBC driver

     rst.Open "SELECT * FROM tblCustomer", _

      "DRIVER=Microsoft Access Driver (*.mdb);" & _

      "DBQ=C:\My Documents\vba.mdb;", _

      adOpenKeyset, adLockPessimistic

    ' Print the first record's values for three fields.

    Debug.Print rst.Fields("CustomerID").Value, _

        rst.Fields("FirstName").Value, _

        rst.Fields("LastName").Value

    rst.Close

End Sub

Referring to Fields in a Recordset

There are several different ways to refer to fields in a Recordset in order to get
their values. Since you may need to work with another programmer’s code,
you should be familiar with all of them.

The example above used this syntax to refer to fields in the Recordset and get
their values:

Debug.Print rst.Fields("CustomerID").Value, _

See
OpenARecordset in
basRecordset

rst.Fields("FirstName").Value, _

rst.Fields("LastName").Value

This syntax returns the value of a particular field within the Fields collection of
a Recordset object. Generically, this syntax is:

rst.Fields("FieldName").Value

Since the Value property is the default property of a field, you can also use:

rst.Fields("FieldName")

And, since the Fields collection is the default collection of a Recordset object,
you can omit it:

rst ("FieldName")

The bang (!) operator is another shortcut used when referring to fields in a
Recordset:

rst!FieldName

or

rst![Field Name]

You must include brackets if the name of the field is more than one word.

The syntax using the bang operator is slightly faster to type, but it has a
disadvantage–you can only use it with a string literal. If you want to use a
variable, you must use the parentheses syntax.

Working with Recordset Methods

Moving through a Recordset

To move through a Recordset, you can use the MoveFirst, MoveNext,
MovePrevious, and MoveLast methods. Each of these does exactly what you
would expect, and makes the appropriate row the current row.

TIP: Attempting to move past the last row, or before the first row, will trigger a

run-time error. Either check before moving to avoid the error, or use error

handling to make sure your code doesn’t fail in cases where there are no

rows, or where you move too far in either direction.

Using EOF and BOF

Once you’ve discovered the EOF and BOF properties, you can loop through all
the rows in a Recordset. For example, you may need to write code that
processes each row in a Recordset. The combination of the EOF property and
the MoveNext method, makes this easy.

The following sub loops through all the rows of a Recordset, printing out the
LastName field for each row. This is a very common use of the methods and
properties we’ve covered so far.

See LoopAll in
basRecordset

Sub LoopAll()

    ' Loop through all the rows in a Recordset.

    Dim rst As ADODB.Recordset

    Set rst = New Recordset

    rst.Open "SELECT * FROM tblCustomer", _

      "DRIVER=Microsoft Access Driver (*.mdb);" & _

      "DBQ=C:\My Documents\vba.mdb;", _

      adOpenKeyset, adLockPessimistic

    Do Until rst.EOF

        Debug.Print rst!LastName

        ' Don't forget this line!

        ' Or you won't go very far.

        rst.MoveNext

    Loop

    rst.Close

End Sub

NOTE As you can see, there’s no point in checking for the existence of
any rows if all you’re going to do is loop through the rows until
EOF. The loop itself will check, and will not execute at all if there
are no rows to be found.

You would be likely to use the BOF property if you had moved to the last row
(MoveLast) and then proceeded to move backwards through the rows
(MovePrevious).

A Simple Example

The following example prompts you for a city name, and then displays a list of
all the customers in that city using a SQL statement to locate all the customers.

See FindByCity in
basRecordset

Sub FindByCity()

    Dim rst As ADODB.Recordset

    Dim strCity As String

    Dim strList As String

    ' Ask for a city name.

    strCity = InputBox( _

     "List customers in what city?" & _

     vbCrLf & "(Try Redmond, Bellevue, or Seattle)")

    ' Make sure there's a city to go for!

    If Len(strCity) <> 0 Then

        Set rst = New Recordset

     rst.Open "SELECT * FROM tblCustomer " & _

            "WHERE [City] = '" & strCity & "'", _

        "DRIVER=Microsoft Access Driver (*.mdb);" & _

        "DBQ=C:\My Documents\vba.mdb;", _

        adOpenKeyset, adLockPessimistic

        ' Build a string of the names in the city

        Do Until rst.EOF

            strList = strList & vbCrLf & rst!LastName

            rst.MoveNext

        Loop

        rst.Close

        MsgBox strList

    End If

End Sub

Summary

• Data access objects enable you to use VBA to work with the data in a
database.

• Use the OpenRecordset method of a database to open a Recordset
based on a table or query.

• Use the Fields collection of a Recordset to work with the fields, or the
data, in a Recordset.

• Loop through a recordset with the MoveNext method.

• Use a SQL statement with a Where clause to get at the data you want
in a recordset.


